

BE TELECOMMUNICATIONS

**TRANSMISSIONS DE DONNEES
NUMERIQUES PAR FAISCEAU LASER
A L'AIR LIBRE**

PREMIERE PARTIE

SOMMAIRE

<u>CHAPITRE 1 : PRESENTATION GENERALE</u>	<u>2</u>
ASPECTS GENERAUX :	2
LE LASER	2
COMMENT COMMANDER LA DIODE ?	3
LA PHOTODIODE	3
LE CODAGE	4
MODULATION / DEMODULATION EN AMPLITUDE :	5
LA CONCEPTION :	5
<u>CHAPITRE 2 : UTILISATION DE SPICE / LE LOGICIEL SMASH.....</u>	<u>6</u>
PREMIER PAS AVEC LE LOGICIEL : CREATION D'UN FILTRE PASSE-BAS.	6
CREATION DE NOTRE FILTRE: REALISATION D'UN FILTRE PASSE-BANDE.	6
COMMENTAIRES SUR LES RESULTATS OBTENUS :	7
UN RESULTAT INSOLUBLE :	8
<u>CHAPITRE 3 : MISE EN PLACE D'UN FILTRE DE BUTTERWORTH 2°ORDRE. ..</u>	<u>8</u>
PREAMBULE :	8
I- CE QUE L'ON CHERCHE A OBTENIR.	8
<u>CHAPITRE 4 : MISE EN PLACE D'UN DEMODULATEUR.</u>	<u>11</u>
PREAMBULE	11
PRINCIPE	11
LE DEMODULATEUR :	12
A- LE CIRCUIT :	12
B- FONCTIONNEMENT DU MONTAGE AMPLIFICATEUR DIFFERENTIEL.	12
C- PLUSIEURS ILLUSTRATIONS SONT ENVISAGEABLES	12
LES MANIPULATIONS EFFECTUEES	13
A- REALISATION LOGICIELLE DU CIRCUIT.	13
B- OBSERVATIONS	14
C- MISE EN PLACE DU DEMODULATEUR A LA SORTIE DU FILTRE DE BUTTERWORTH.	15
D- MISE EN PLACE D'UN FILTRE PASSE BAS.	15
<u>CHAPITRE 5: REALISATION MATERIELLE</u>	<u>16</u>
MISE EN PLACE DU FILTRE DE BUTTERWORTH	16
MISE EN PLACE DU DEMODULATEUR	17

Chapitre 1 : Présentation générale

Aspects généraux :



Le faisceau devra porter un message pour distinguer qui a atteint le récepteur. On aura aussi un rayon continu (pointeur) caractérisé par une diode.

On veut que l'émetteur et le récepteur soient mobiles. Donc, il faudra que l'alimentation soit portable et légère, que l'autonomie pour la durée du jeu ne soit pas trop courte et il faudra prendre en compte le poids du matériel.

De plus, il nous faudra un système non dangereux (en l'occurrence pour les yeux). Donc, il faudra limiter la puissance du Laser à 1mW ($P_{\text{Laser}} < 1\text{mW}$).


Enfin, le débit du message devra être de 4 kbits/s.

Le Laser



$$Pr = Pe$$

$$RSB = \frac{Pe}{DbSr}$$



$$Pr = Pe \frac{Sr}{Se}$$

$$RSB = \frac{Pe}{DbSe}$$

Comment commander la diode ?


Il existe deux approches possibles pour la commande de la diode :

Commande en tension : la diode peut brûler. Elle n'est donc pas stable. En effet, si le courant augmente légèrement, la température augmente (cf graphes ci-dessous) ce qui entretient l'augmentation du courant. Ainsi, le courant ne cesse d'augmenter jusqu'à brûler le composant.

Commande en courant : Si le courant augmente légèrement, la température aussi. Mais dans ce cas là, la puissance diminue entraînant une chute légère en température. Le système reste alors stable car il s'autorégule.


On commandera donc la diode en courant.


Remarque : au final, on veut une puissance de 1mW. Or il existe une dispersion dans les composants et qui dit dispersion, dit variation de courant ΔI qui entraîne une variation de puissance ΔP . Donc il faut un régulateur de puissance pour $P_{optique}$.



La photodiode


Lorsque la photodiode est polarisée en direct, elle correspond à une diode classique. En inverse, elle correspond à une diode photosensible. On polarisera donc la diode en inverse. On aura besoin d'un filtre pour filtrer les bruits.





Le codage

Le signal à transmettre étant un signal numérique formé de '0' et de '1' logiques modulant le faisceau laser. Le codage de ces valeurs logiques se fera par un format de modulation de largeur :
Le symbole '0' sera un signal qui reste à '1' pendant 1/3 de la période et '0' pendant 2/3 de la période.
Le symbole '1' sera un signal qui reste à '1' pendant 2/3 de la période et '0' pendant 1/3 de la période.




Modulation / Démodulation en amplitude :

Pour pouvoir filtrer, nous allons adopter une méthode vue en TD, en multipliant le signal par un signal sinusoïdal pur :

Comme on fait varier l'amplitude, on effectue alors une modulation en amplitude pour la porteuse.

Après le codage, on pourra discerner l'enveloppe du signal. Le but du récepteur sera de détecter cette enveloppe. (C'est la démodulation) :



La conception :

Au niveau de l'émetteur :

L'émetteur comprend une diode laser, un régulateur ($P=1\text{mW}$) et un microcontrôleur qui fournit le signal à transmettre modulé à 100Hz.

Il faut générer un signal sinusoïdal de période 10 microseconde ($F=100\text{Hz}$). Ce que l'on va faire mais qui est inacceptable en télécommunication, c'est générer un signal carré à l'aide du microcontrôleur PIC 16F84A.

Au niveau du récepteur :

On a une photodiode suivie du récepteur.

Le récepteur comprendra tout d'abord un filtre passe-bande réglé autour de 100kHz avec une bande passante de 10 à 15kHz. La sortie de ce filtre sera ensuite envoyée dans un étage de démodulation qui consistera en un redressement et un filtrage passe-bas. Enfin, un comparateur permettra de régénérer le signal (remise en forme) avant de le faire passer dans un PIC 16F84A qui se chargera du décodage.


Remarque : lorsque l'on fait un filtre, on travaille en tension.

Chapitre 2 : utilisation de SPICE / le logiciel SMASH.

Le logiciel SMASH est un simulateur SPICE permettant d'évaluer le comportement de circuits aussi bien en régime temporel qu'en régime fréquentiel. La résolution numérique du système par le logiciel permet de visualiser le comportement théorique d'un circuit avant de passer à sa réalisation pratique.


Premier pas avec le logiciel : création d'un filtre passe-bas.

Le schéma du filtre est :



Le listing entré dans SMASH est le suivant :

```
* Impédances du circuit.
R1 1 2 159
C1 0 2 10n
* Générateurs.
Vin 1 0 AC 1
* Commandes
.AC DEC 100 1k 1g
.TRACE AC VDB(2)
.TRACE TRAN V(1)
.TRACE TRAN V(2)
.TRACE TRAN V(2)
```




Remarque : afin d'obtenir une fréquence de 100Hz, on se fixe $C=10\text{nF}$, et comme la fréquence de coupure est : $F_c = \frac{1}{2\pi RC}$ alors on déduit la résistance à mettre : $R = \frac{1}{2\pi F_c C} = 159\Omega$.

Création de notre filtre: Réalisation d'un filtre Passe-Bande.


Dans l'étage de réception de notre projet, il y a un filtre passe bande permettant de ne garder que les composantes spectrales du signal utile, avant de passer à une réalisation pratique de celui-ci, nous l'avons simulé à l'aide du logiciel SMASH.

Une première version de ce filtre consiste à placer en cascade un filtre passe bas et un filtre passe haut :




Afin d'adapter les impédances et d'éviter d'avoir les deux condensateurs en liaison directe, on applique des amplificateur opérationnels (AOP). Ceci permet d'éviter les interactions entre les deux étages du circuit.

Voici alors le schéma fonctionnel du filtre à réaliser :



Pour décrire les AOP dans le logiciel SMASH, on a besoin de bibliothèques et de spécifications précises mises en place par le biais de sous circuits et de bibliothèques.


Schéma de l'AOP :




Voir le listing du filtre passe Bande : annexe 1

Commentaires sur les résultats obtenus :

On a effectué un filtre Passe Bande et on a la courbe suivante :



Le problème constaté est que la bande passante est trop élevée et que l'on diminue le gain. (-6db !). En effet, on a les recouvrements qualitatifs suivants :




Un résultat insoluble :

En effet, pour avoir un gain à 0db (ce que l'on souhaite), il faudrait repousser la courbe rouge à gauche et la bleue à droite. C'est à dire :

Diminuer $F_{cPasseHaut}$

Augmenter $F_{cPasseBas}$

Cependant, si on fait cela, on obtient :



On augmente la Bande Passante. On ne peut donc pas avoir les deux contraintes simultanément, à savoir une bande passante étroite (10 à 15 KHz) et un gain nul. Le problème constaté n'est pas soluble avec cette architecture de filtre.

Nous allons alors utiliser une autre méthode.


Chapitre 3 : Mise en place d'un filtre de ButterWorth 2° ordre.

Préambule :


Nous avons essayé auparavant de mettre en place un filtre passif du premier ordre. Nous allons ici nous intéresser à la mise en place d'un filtre actif du second ordre.

I- Ce que l'on cherche à obtenir.

On veut la caractéristique suivante :



Le filtre est donc représenté par ce circuit :



La fonction de transfert est la suivante :

$$H(p) = -\frac{\frac{p}{R1C}}{p^2 + \frac{2}{R3C}p + \left(\frac{1}{R1} + \frac{1}{R2}\right)\frac{1}{R3C^2}} = -\frac{K.\Delta\omega.p}{p^2 + \Delta\omega.p + \omega_0^2}$$

avec $\omega_0 = 2\pi f_0$; $\Delta\omega = 2\pi\Delta f$;

On souhaite approcher au mieux $f_0 = 100$ kHz, $\Delta f = 10$ KHz, et $K=1$.
On effectue alors les calculs en se fixant une valeur pour la capacité.
On assimile les coefficients et il résulte que :

$R3 = 2 * R1$
$R3 = \frac{1}{\pi C \Delta f}$
$R2 = \frac{1}{R3C^2 (2\pi f_0)^2 - \frac{1}{R1}}$


Manipulations et déroulement du TP

Suite à la première séance, nous avons vu que le filtre n'était pas adapté. Nous avons donc utilisé un filtre ButterWorth définie tel que nous l'avons présenté ci-dessus.

Dans un premier temps, nous avons fixé une valeur existante de capacité et déduit des résistances plausibles nous permettant d'aboutir aux valeurs désirés, à savoir $f_0 = 100$ KHz, $\Delta f = 10$ KHz, $K = 1$.

Bien entendu, il faut tenir compte des valeurs des impédances disponibles matériellement. Pour cela, nous avons utilisé le logiciel Smash en configurant le circuit et en simulant à plusieurs reprises.

Dans un premier temps, nous obtenions une courbe décalée :




Mais nous avons quand même un gain raisonnable légèrement inférieur à 0db. Il nous a alors fallu régler les résistances un peu par tâtonnement et par doigté de façon qualitative. Finalement, nous avons fixé des valeurs très suffisantes :

$$\begin{aligned} R1 &= 14000 \\ R2 &= 85 \\ R3 = R4 &= 27000 \\ C &= 1 \text{ nF} \end{aligned}$$

Avec ces valeurs, la courbe de réponse en fréquence du filtre est la suivante :



La réponse du filtre lorsque l'on applique un signal sinusoïdal à 100kHz en entrée est la suivante :



On voit que le filtre laisse bien passer le signal. Cependant, l'amplitude a beaucoup diminuée : de 5V à 1V !

On a un plan de BODE très correcte mais on se rend compte quand on teste le filtre (voir listings ci dessous), que l'amplitude diminue en sortie. De plus on a un régime transitoire au début avec des oscillations. On ne sait pas trop d'où cela peut venir, mais on pense que cela doit être au niveau des résistances. Donc, on a essayé d'ajuster les résistances afin d'obtenir une courbe correcte.

Ainsi, on se rend compte de l'utilité du logiciel Smash qui est un bon moyen pour tester notre circuit en changeant à loisir nos valeurs de résistances avant de passer à l'exécution purement physique.

Remarque : Suite à la deuxième séance, nous avons remarqué que le gain en tension de notre filtre ButterWorth était diminué. En effet, l'AOP ne génère pas assez de courant pour avoir 5V en sortie. On placera donc une résistance faible en sortie (1k à 10k) pour obtenir une sortie « normale ».

Voir annexe 2 : Listing du filtre de BUTTERWORTH


Chapitre 4 : Mise en place d'un démodulateur.

Préambule


Au cours de cette troisième séance, on a mis en place un démodulateur que l'on placera à la sortie de notre filtre de butterworth.

Principe

A partir de notre signal, il va nous falloir effectuer la démodulation, c'est à dire détecter l'enveloppe du signal.



Si on faisait passer le signal a dans un circuit spécifique de diodes, on ne garderait seulement soit les tensions positives, soit les tensions négatives. Afin de garder aussi bien les alternances positives que les alternances négatives du signal, nous allons utiliser deux circuits à diodes et à amplificateurs opérationnels chargés chacun de récupérer l'une de ces deux alternances, nous ferons ensuite la somme de ces deux signaux affectés des signes permettant d'obtenir un redressement double alternance de la tension en entrée comme détaillé ci-dessous :



Ainsi, pour notre signal modulé, après filtrage par le filtre de butterworth et redressement, on obtient :




C'est cette méthode que l'on va utiliser. Afin de ne garder que l'enveloppe, il est nécessaire d'éliminer les variations rapides donc de garder les basses fréquences en atténuant les hautes fréquences. Nous allons utiliser un filtre passe bas qui va permettre de « smoother » (lisser) la courbe.

Remarque : on peut avoir de faibles oscillations au niveau de la courbe, et on pourrait effectuer une détection de seuil par trigger de Schmit pour régénérer le signal.

Le démodulateur :

A- Le Circuit :




B- Fonctionnement du montage amplificateur différentiel.

On peut établir les formules suivantes de V_+ et V_- en utilisant la méthode du pont diviseur de tension.

$$V_+ = \frac{R_{60}}{R_{40} + R_{60}} V_2$$

$$V_- = \frac{R_{50}}{R_{30} + R_{50}} V_1 + \frac{R_{30}}{R_{30} + R_{50}} V_S$$



On a fait l'hypothèse d'un AOP parfait : $V_+ = V_-$.

On obtient donc :

$$V_S = \frac{(R_{20} + R_{50})R_{60}}{(R_{40} + R_{60})R_{30}} V_2 - \frac{R_{50}}{R_{30}} V_1 \quad \text{et} \quad V_S = V_2 - V_1$$

Cependant, ce résultat n'est valable que lorsque le circuit est utilisé séparément (circuit isolé).


En effet, si R_{10} et R_{30} ont les mêmes valeurs, le courant dans R_{50} peut se propager et modifier V_1 : il y a des fuites. A cause des impédances, on aura une autre expression de V_S qui n'est pas la même qu'en circuit isolé !

C- Plusieurs illustrations sont envisageables

Afin de déterminer les impédances, il est nécessaire de regarder le comportement du circuit en fonction de la tension V_e en entrée.

Le signal $V_e > 0$:

On a donc $V_2 = V_e$ et « $V_1 = 0$ ».



$$V_+ = \frac{R_{60}}{R_{40} + R_{60}} V_2$$


$$V_- = \frac{R_{10} + R_{30}}{R_{10} + R_{30} + R_{50}} V_S$$

d'où :
$$V_{S(0)} = \frac{R_{60}(R_{10} + R_{30} + R_{50})}{(R_{40} + R_{60})(R_{10} + R_{30})} V_2$$

En fait, « $V_1 = 0$ », puisque la diode bloquant le courant, on ne considère pas la partie gauche qui engendre un potentiel nul en V_1 . C'est pour cela que l'on considère les deux résistances R_1 et R_{30} .

Le signal $V_e < 0$:

On a donc $V_1 = V_e$ et « $V_c = 0$ ».



$$V^+ = 0$$

$$V_- = \frac{R_{50}}{R_{30} + R_{50}} V_1 + \frac{R_{30}}{R_{50} + R_{30}} V_s$$

$$V_{S(b)} = -\frac{R_{50}}{R_{30}} V_1$$

$$V_S = \frac{R_{60}(R_{30} + R_{50} + R_{10})}{(R_{40} + R_{60})(R_{30} + R_{10})} V_2 - \frac{R_{50}}{R_{30}} V_1$$

Comme on veut $V_S = V_2 - V_1$ on en déduit qu'il faut prendre $R = R_{10} = R_{20} = R_{30} = R_{40} = R_{50}$ et $R_{60} = 2R$.

Remarque : pour réaliser le circuit dans SMASH, on introduit un nouvel élément que l'on n'utilisait pas avant : la diode. Sa déclaration est la suivante :

.Model 1N4148 D(IS=0.1p Rs=16 CJO=2p Tt=12n Bv=100 Ibv=0.1p)
Diode n_A n_k 1N4148

Avec A pour Anode et K pour cathode.

Les manipulations effectuées


A- Réalisation logicielle du circuit.

On définit le circuit de démodulation sur le logiciel SMASH. (*voir listing annexe*). On teste au fur et à mesure de la construction « logicielle » du circuit le bon fonctionnement des différents éléments. C'est à dire que l'on définit d'abord des sous parties comme le montage suiveur avec la diode placé en sortie afin de visualiser si le résultat est correcte (suiveur de tension donc gain en sortie de 1 en théorie, puisque en fait, l'amplitude diminue légèrement) et blocage des tensions soit positives, soit négatives selon le sens de la diode.


Cela évite d'avoir à « déboguer » le circuit une fois qu'il est complet, s'il y a des erreurs il est plus simple de les corriger avant d'avoir tout monté qu'une fois que tout est écrit.

B- Observations

On remarque des saturations en utilisant un signal sinusoïdal d'amplitude 5V en entrée. Le résultat est cohérent avec ce que l'on attendait mais SMASH détecte des saturations qu'il nous signale sous forme de messages d'erreurs. Les courbes obtenues dans un premier temps sont les suivantes :



Sortie du démodulateur



V(1) : signal d'entrée du démodulateur

V(7) : signal de la sortie de l' AOP suiveur de tension (diode polarisée directe)

V(2) : signal à la sortie de l' AOP suiveur de tension. (diode polarisée inverse)

En fait, on pense que cela est dû à la rétroaction dans le circuit qui demande à l'AOP de fournir plus de tension, d'où le besoin de forte impédance signalant un message d'erreur.

Ensuite, on a testé un signal sinusoïdal en entrée d'amplitude 3V et on a pu constater que SMASH ne signalait plus d'erreurs de résolution. L'amplitude du signal envoyé en entrée avait donc une importance. Or, le signal que l'on fournira ne pourra pas être à 5V, mais en dessous, ce qui nous arrange.

Remarque : L'effet de rétroaction : on constate qu'en observant au point 2 (V(2)), on a pas exactement une tension de 0 à -4V mais plutôt une tension de 1V à -4V : ceci est dû au fait qu'en

V(1) (vu théoriquement précédemment), on a pas exactement $V_1=0V$!!! On a des fuites de rétroaction qui proviennent du dernier AOP branché dans le circuit.

C- Mise en place du démodulateur à la sortie du filtre de ButterWorth.

Une fois le démodulateur simulé de façon satisfaisante, nous avons testé le circuit en connectant en entrée la sortie du filtre de ButterWorth.

Au niveau programmation de SMASH, nous avons réalisé des sous circuits pour l'ensemble du démodulateur (trois paramètres à prendre en compte : entrée, sortie et tension d'alimentation des AOP) ainsi que l'ensemble du filtre de ButterWorth. Cela nous permet d'avoir un code plus modulable et plus clair. (Voir listing annexe)

Nous avons effectué le test avec une source « source modulée » en entrée qui simule les variations du signal laser que l'on aura en entrée. Nous avons d'abord obtenu des oscillations avec une faible amplitude.

Après avoir modifié la fréquence de la porteuse et on rajouté du bruit au signal, nous avons obtenu des courbes très appréciables :

Ainsi, nous avons testé le filtre ButterWorth en haute et basse fréquence afin de vérifier sa bonne intégration dans la chaîne de réception.

Remarque : Dans la version proposée de SMASH, il existe une limite à un certain nombre de nœuds. Afin de palier à cette limitation, nous avons dû créer des sous circuits qui nous ont permis de réduire le nombre total de nœuds.

D- Mise en place d'un filtre passe bas.

Après avoir mis en place le montage permettant de redresser notre signal, nous avons mis en place un filtre passe bas afin de « lisser la courbe », c'est à dire obtenir l'enveloppe de la courbe en éliminant les oscillations hautes fréquences.

La réponse doit être telle que l'on pourra distinguer le mieux possible les seuils à atteindre lorsque l'on déterminera l'enveloppe du signal démodulé.

A l'aide du logiciel, nous avons déterminé nos valeurs de résistance et capacité à mettre en place afin de ne laisser que les basses fréquences du signal.

Pour une résistance de $R = 1,5 \text{ kohm}$ et une capacité $C=10\text{nF}$ la fréquence de coupure vaut 10kHz ce qui permet de bien lisser la courbe sans trop perdre en amplitude.

Remarque : Nous constatons que l'amplitude est diminuée, et que la courbe ne suit pas exactement l'enveloppe du signal en entrée du filtre.

En effet, on utilise ici un filtre RC comme filtre passe bas, qui est un filtre passif. On ne peut pas garder la même amplitude.

Pour pouvoir suivre la courbe et obtenir une réponse plus nette, on pourrait utiliser un filtre actif.

Chapitre 5: Réalisation Matérielle

Mise en place du filtre de ButterWorth

Cette partie est essentiellement pratique, il s'agit du câblage des montages précédemment simulés à l'aide du logiciel SMASH. Nous avons choisi nos composants en tenant compte des marges d'erreurs sur les valeurs des impédances qui sont de $\pm 5\%$

Pour le branchement de l'AOP, il nous a été utile de consulter la page de documentation fournie afin de savoir comment l'alimenter et de savoir où devaient s'effectuer les branchements. Nous avons aussi dû prendre conscience de la mise en place des éléments sur la plaque, en respectant les potentiels, et en coupant les fils de telle manière à rendre le circuit plus lisible et plus facile à analyser.

Une fois les composants mis en place et les branchements effectués, nous avons réglé notre signal sinusoïdal délivré par le générateur de fonctions à l'oscilloscope avant de l'envoyer en entrée. Nous l'avons réglé sur 3V d'amplitude soit 6V crête à crête.

Ensuite nous avons analysé les potentiels en différents points du circuit afin de vérifier d'une part les tensions envoyées en entrée, et les divers potentiels du circuit. Puis nous avons analysé notre signal en sortie de l'AOP à l'oscilloscope.

Nous avons été assez surpris du résultat, en effet, il ne correspondait pas en tout point avec ce que l'on attendait. Bien qu'aux basses fréquences, le signal était significativement atténué, le maximum de gain se trouvait à la fréquence de 115 kHz. Nous souhaitions un maximum de gain à 100 kHz.

Nous avons alors cherché d'autres valeurs pour avoir la caractéristique recherchées pour notre filtre. Constatant un décalage par rapport à la simulation, nous en avons tenu compte lors d'une nouvelle simulation, ainsi nous avons modifié nos valeurs :

$$\begin{aligned}R_1 &= 13000 \text{ ohm} \\R_2 &= 110 \text{ ohm} \\R_3 &= 27000 \text{ ohm} \\C &= 1 \text{ nF}\end{aligned}$$

Avec ces valeurs, nous avons une simulation pour laquelle le maximum de gain se situait à 90kHz avec une largeur de bande de 12kHz, après réalisation et mesure, nous avons trouvé un maximum de gain pour la fréquence de 101kHz et une largeur de bande (à -3db) voisine de 12kHz.

Remarque :

On peut sur l'oscilloscope déterminer le gain obtenu en sortie de l'AOP.

$G = V_{\text{sortie}}/V_{\text{entrée}}$ et en dB on a $20\log(G)$. Dans notre cas, on a un gain de -2db.

Cette valeur n'est pas la plus satisfaisante car on devrait avoir 0dB.

Mais cela n'aura de grande répercussion étant donné que l'on est bien centré sur 100 kHz.


Pour déterminer la bande passante, il suffit de savoir que $G_{db} - 3db = \frac{V_{out}}{\sqrt{2}} = 4,55V$ dans notre cas.

Puis on fait varier la fréquence autour de 100kHz jusqu'à ce qu'on arrive à $V_{out} = 4,55V$. On a alors F_{c1} et F_{c2} et $BP = |F_{c1} - F_{c2}|$.

Mise en place du démodulateur

On a choisi : $R_{10}=R_{20}=R_{30}=R_{40}=R_{50}=R= 10 \text{ kohm}$ et $R_{60}=2R=20\text{kohm}$.

Pour le branchement avec les AOP, on a eu recours à sa fiche caractéristique :



ANNEXES : LES LISTINGS DE NOS PROGRAMMES

ANNEXE 1 : listing du filtre Passe Bande

*nécessité des librairies pour l'AOP

.lib ZA_AOP.AMD

.Lib ZA_FOLLOWER.AMD

*Définition du sous circuit AOP

.SUBCKT AOP2604 EP EM VP VM OUT

ZOPAMP

+IN(EP EM EP EM OUT VP VM)

+OUT(X/ZS=1E6,1E-10)

+PAR(10000 10E6)

+ZA_AOP

ZVFOLLOW

+IN(X)

+OUT(OUT/ZS=100,0.1P)

+PAR(1)

+ZA_FOLLOWER

.ENDS

*Création Du filtre passe bas :

* Impédances du circuit.

R1 1 2 159

C1 0 2 10n

* Générateurs.

Vin 1 0 AC 1

* Commandes

.AC DEC 100 1k 1g

.TRACE AC VDB(2) MIN=-1.9000000E+002 MAX=1.9000000E+002

.TRACE AC VDB(5) MIN=-8.4967474E+000 MAX=-5.5076410E+000

.TRACE TRAN V(2) MIN=-1.2000000E+000 MAX=1.2000000E+000

*Premier AOP

XAOP1 2 3 6 7 3 AOP2604

*Céation du filtre passe haut

*Impédances :

R2 0 4 159

C2 3 4 10n

*Commandes

.AC DEC 100 1K 1G

*Deuxieme AOP

XAOP2 4 5 6 7 5 AOP2604

*Alim AOP

Vplus 6 0 5

Vmoins 7 0 -5

ANNEXE2 : Listing du filtre de BUTTERWORTH

```
*****  
***** BUTTERWORTH ordre 2 - Passe Bande *****  
*****
```

*nécessité des bibliothèques pour l'AOP

.lib ZA_AOP.AMD

.Lib ZA_FOLLOWER.AMD

*Définition du sous circuit AOP

.SUBCKT AOP2604 EP EM VP VM OUT

ZOPAMP

+IN(EP EM EP EM OUT VP VM)

+OUT(X/ZS=1E6,1E-10)

+PAR(10000 10E6)

+ZA_AOP

ZVFOLLOW

+IN(X)

+OUT(OUT/ZS=100,0.1P)

+PAR(1)

+ZA_FOLLOWER

.ENDS

* Sous circuit générateur de signaux de test type lasertrucmuche.

.SUBCKT SOURCEMOD OUT

Vsin porteuse 0 sin 0 5 80k

Vdat data 0 pulse 0 1 120u 5u 5u 120u 250u

Emod OUT 0 value{V(data)*V(porteuse)}

.ENDS

* Valeur des composants

R1 1 2 15000

R2 2 0 85

R3 4 0 23000

R4 3 5 23000

C1 2 5 1E-9

C2 2 3 1E-9

* AOP et autres modules

XAOP1 4 3 6 7 5 AOP2604

Xmod 1 sourcemod

* Générateurs

** Décommenter pour analyse de BODE

*Vc 1 0 sin 0 5 100k AC 1

Vplus 6 0 5

Vmoins 7 0 -5

* Commandes

.TRAN 10n 1000u

.plot TRAN V(5)

** Commentés pour voir le transitoire

.AC DEC 100 60K 200K

.plot AC Vdb(5)

Annexe3 : listings du démodulateur

```
*****  
***** Demodulateur*****  
*****
```

*nécessité des bibliothèques pour l'AOP

.lib ZA_AOP.AMD

.Lib ZA_FOLLOWER.AMD

*Définition du sous circuit AOP

.SUBCKT AOP2604 EP EM VP VM OUT

ZOPAMP

+IN(EP EM EP EM OUT VP VM)

+OUT(X/ZS=1E6,1E-10)

+PAR(10000 10E6)

+ZA_AOP

ZVFOLLOW

+IN(X)

+OUT(OUT/ZS=100,0.1P)

+PAR(1)

+ZA_FOLLOWER

.ENDS

*Demodulation

.SUBCKT DEMOD E S VP VM

* Définition de la diode.

.model 1N4148 D(Is=0.1p Rs=16 CJO=2p Tt=12n Bv=100 Ibv=0.1p)

XAOP2 E 7 VP VM 8 AOP2604

Diode2 8 7 1N4148

R20 7 0 10k

XAOP1 E 2 VP VM 3 AOP2604

Diode1 2 3 1N4148

R10 2 0 10k

R30 2 4 10k

R40 7 6 10k

R60 6 0 20k

XAOP3 6 4 VP VM S AOP2604

R50 4 S 10k

.ENDS

XDEM 1 5 9 10 DEMOD

* Générateurs

Vc 1 0 sin 0 3 100k AC 1

Vplus 9 0 5

Vmoins 10 0 -5

.TRAN 10n 100u

.Plot TRAN V(1)

.PLOT TRAN V(5)

ANNEXE4 : Listings du démodulateur + filtre de Butterworth

```
*****  
***** Demodulateur + Butterworth *****  
*****
```

*nécessité des bibliothèques pour l'AOP

.lib ZA_AOP.AMD

.Lib ZA_FOLLOWER.AMD

*Définition du sous circuit AOP

.SUBCKT AOP2604 EP EM VP VM OUT

ZOPAMP

+IN(EP EM EP EM OUT VP VM)

+OUT(X/ZS=1E6,1E-10)

+PAR(10000 10E6)

+ZA_AOP

ZVFOLLOW

+IN(X)

+OUT(OUT/ZS=100,0.1P)

+PAR(1)

+ZA_FOLLOWER

.ENDS

*Demodulation

.SUBCKT DEMOD E S VP VM

* Définition de la diode.

.model 1N4148 D(Is=0.1p Rs=16 CJO=2p Tt=12n Bv=100 Ibv=0.1p)

XAOP2 E 7 VP VM 8 AOP2604

Diode2 8 7 1N4148

R20 7 0 10k

XAOP1 E 2 VP VM 3 AOP2604
Diode1 2 3 1N4148
R10 2 0 10k

R30 2 4 10k
R40 7 6 10k
R60 6 0 20k

XAOP3 6 4 VP VM S AOP2604
R50 4 S 10k

.ENDS

*Butterworth
.SUBCKT BUTTERWORTH E S VP VM

* Valeur des composants

R1 E 2 14000
R2 2 0 85
R3 4 0 27000
R4 3 S 27000
C1 2 S 1E-9
C2 2 3 1E-9

* AOP et autres modules

XAOP1 4 3 VP VM S AOP2604

.ENDS

* Sous circuit générateur de signaux de test type lasertrucmuche.

.SUBCKT SOURCEMOD OUT

Vsin porteuse 0 sin 0 5 100k
Vdat data 0 pulse 0 1 120u 5u 5u 120u 250u
*VBruit bruit 0 sin 0 1 5k
*VBTHF bruithf 0 sin 0 1 1meg
Emod OUT 0 value{V(data)*V(porteuse)}

.ENDS

XDEM 2 5 9 10 DEMOD
XBUTTERWORTH 1 2 9 10 BUTTERWORTH
Xmod 1 sourcemod

* Générateurs

*Vc 1 0 sin 0 3 100k AC 1
Vplus 9 0 5
Vmoins 10 0 -5

.TRAN 10n 1000u
.Plot TRAN V(1)
.Plot TRAN V(2)
.PLOT TRAN V(5)